Compressibility Factor from Redlick-Kwong Equations

(Dr. Tom Co 9/2/08)

Working Equations:

(based on Cutlip and Shacham, 2008, pp. 101-103)

Let P be pressure in atm, T be temperature in K and $\hat{V}$ be molar volume in $\frac{liters}{g-mol}$. The Redlich-Kwong equation is given by

$$P = \frac{RT}{\hat{V} - b} - \frac{a}{\hat{V}(\hat{V} + b)\sqrt{T}} \tag{1}$$

where

$$a = 0.42747\left(\frac{R^2 T_c^{\frac{5}{2}}}{P_c}\right) \tag{2}$$

$$b = 0.08664\left(\frac{RT_c}{P_c}\right) \tag{3}$$

Suppose we want to obtain compressibility factor

$$z = \frac{P\hat{V}}{RT} \tag{4}$$

as a function reduced pressure $P_r = P/P_c$, at various cases of reduced temperature $T_r = T/T_c$.

First, solve for $\hat{V}$ in (4),

$$\hat{V} = \frac{zRT}{P} \tag{5}$$

then substitute (5) in (1) to obtain a cubic equation in z given by

$$z^3 - z^2 - qz - r \tag{6}$$

where,

$$r = AB \tag{7}$$

$$q = B^2 + B - A \tag{8}$$

$$A = 0.42747\left(\frac{P_r}{T_r^{\frac{5}{2}}}\right) \tag{9}$$

$$B = 0.08664 \left(\frac{P_r}{T_r}\right) \tag{10}$$

If we wish to obtain the compressibility factor of the vapor phase, we need the maximum real-valued root of the cubic equation.

The mcroot Function:

The following code is a function to obtain the maximum real root of a cubic equation:

```
Function mcroot(a3, a2, a1, a0)
'
'   Computes the maximum real root of the cubic equation
'                   a3 x^3 + a2 x^2 + a1 x + a0 = 0
'
    Dim A, B, C, D, z
    A = a2 / a3
    B = a1 / a3
    C = a0 / a3
    p = (-A ^ 2 / 3 + B) / 3
    q = (9 * A * B - 2 * A ^ 3 - 27 * C) / 54
    Disc = q ^ 2 + p ^ 3
    If Disc > 0 Then
        h = q + Disc ^ (1 / 2)
        y = (Abs(h)) ^ (1 / 3)
        If h < 0 Then y = -y
        z = y - p / y - A / 3
    Else
        theta = Atn((-Disc) ^ (1 / 2) / q)
        c1 = Cos(theta / 3)
        If q < 0 Then
            s1 = sin(theta / 3)
            c1 = (c1 - s1 * 3 ^ (1 / 2)) / 2
        End If
        z1 = 2 * (-p) ^ (1 / 2) * c1 - A / 3
        m = A + z1
        r = (m ^ 2 - 4 * (B + m * z1)) ^ (1 / 2)
        z2 = (-m + r) / 2
        z3 = (-m - r) / 2
        z = z1
        If z2 > z Then z = z2
        If z3 > z Then z = z3
    End If
    mcroot = z
End Function
```

Figure 1. **mcroot** Code.

To include the function in an Excel worksheet:

1. Open the worksheet.
2. Press **[Alt-F11]** to open the VBA editor.
3. Click on the module (if it does not exist click **[Insert]→[Module]** to create).
4. Copy (or cut-and-paste) the function code above into the code window.
5. Press **[Alt-F11]** once more to go back to Excel worksheet.
6. Test the function.

Example: Compressibility of Steam for $P_r = 0.1, 0.2, \ldots, 10$ at $T_r = 1, 1.2, 1.5, 2, 3$.

	A	B	C	D	E	F	G	H
1	Component		Steam					
2		R	0.08206					
3		Tc	647					
4		Pc	218					
5		Pr	1.2					
6		Tr	1					
7		A	0.512964					
8		B	0.103968					
9		q	-0.39819					
10		r	0.053332					
11		z	0.25788					
12								
13				Tr=1	Tr=1.2	Tr=1.5	Tr=2	Tr=3
14			0.25788	1	1.2	1.5	2	3
15			0.1	0.965162	0.979972	0.990293	0.996817	1.000162
16			0.2	0.928637	0.959637	0.980652	0.993718	1.000356
110			9.6	1.206428	1.137806	1.107138	1.118883	1.136462
111			9.7	1.216871	1.146476	1.113608	1.122948	1.138788
112			9.8	1.227301	1.155138	1.120084	1.127031	1.141125
113			9.9	1.237718	1.163792	1.126568	1.131134	1.143473
114			10	1.248122	1.172438	1.133057	1.135255	1.145832

=mcroot(1,-1,-C9,-C10)

Figure 2. Data table for compressibility factors.



Figure 3. Compressibility chart.